Függvénydiszkusszió

Ebben a fejezetben áttekintjük és helyenként tovább finomítjuk a függvényvizsgálat eszközrendszerét. Ezután példákat mutatunk a teljes függvényvizsgálatra.

A függvényvizsgálat szempontjai

Értelmezési tartomány

A függvény megadásakor kétféleképpen is eljárhatunk.:
1) A függvényt az értelmezési tartománnyal együtt adjuk meg ,vagy valamilyen jelenség megfigyelésekor jön létre a függvény, amely az utóbbi esetben véges számú értékpárból áll. Ezekben az esetekben az értelmezési tartomány vizsgálata nem jön szóba.
2) A függvényt generáló utasítást adunk meg csupán. Ilyenkor mintegy "gépnek" tekintjük a függvényt, amely kiszámítja a bemeneti értékhez tartozó függvényértéket:

Ilyenkor feladat lehet az értelmezési tartomány vizsgálata.A kérdés úgy vetõdhet föl, hogy határozzuk meg egy adott halmaz (az sokszor a valós számok **R** jelû halmaza) legbõvebb olyan részhalmazát, amelyen a szóbanforgó utasítással függvény hozható létre.

Példa.

Legyen adott mért értékek két adatsora. A független változó értékei:

```
> X:=[1/4, 1/2, 3/4, 1, 5/4, 3/2, 7/4, 2, 9/4, 5/2, 11/4, 3, 13/4, 7/2, 15/4, 4];
```

$$X := \left[\frac{1}{4}, \frac{1}{2}, \frac{3}{4}, 1, \frac{5}{4}, \frac{3}{2}, \frac{7}{4}, 2, \frac{9}{4}, \frac{5}{2}, \frac{11}{4}, 3, \frac{13}{4}, \frac{7}{2}, \frac{15}{4}, 4\right] \quad (1.1.1)$$

A megfelelõ függvényértékek:

```
> Y:=[-2.468, -1.587, -1.126, -.7148, -.4858, -.2662, -.07670, .2329, .3243, .4907, .5865, .6667, .8094, .8844, .9189, 1.062]
;
```

$$Y := [-2.468, -1.587, -1.126, -.7148, -.4858, -.2662, -0.07670, 0.2329, 0.3243, 0.4907, 0.5865, 0.6667, 0.8094, 0.8844, 0.9189, 1.062] \quad (1.1.2)$$

Készítsük el a zip utasítással az összetartozó értékpárokat. Természetesen sokszor inkább ezek adottak, a mérés nyomán keletkezõ jegyzõkönyvben, de mi most be szeretnénk mutatni a két adatsor egyesítését is.

```
> pontok:=zip((x,y)->[x,y],X,Y);
```

$$pontok := \left[\left[\frac{1}{4}, -2.468\right], \left[\frac{1}{2}, -1.587\right], \left[\frac{3}{4}, -1.126\right], [1, -.7148], \left[\frac{5}{4}, -.4858\right], \left[\frac{3}{2}, -.2662\right], \left[\frac{7}{4}, -0.07670\right], [2, 0.2329], \left[\frac{9}{4}, 0.3243\right], \left[\frac{5}{2}, 0.4907\right], \left[\frac{11}{4}, \right.\right. \quad (1.1.3)$$

0.5865], [3, 0.6667], $\left[\frac{13}{4}, 0.8094\right]$, $\left[\frac{7}{2}, 0.8844\right]$, $\left[\frac{15}{4}, 0.9189\right]$, [4, 1.062]]

Ezután ábrázolhatjuk az értékpároknak megfelelõ pontokat:

```
> plot(pontok,style=point,color=red,symbol=circle,title=`Az adatsor`);
```



Ebben az esetben az lehet a feladat, hogy olyan képlettel kezelhetõ függvényt határozznk meg, amelyhez "jól ileszkednek" a tapasztalat által szolgáltatott pontok. De ez már egy másik kurzus feladata lehet. Mindenesetre, olyan megérzésünk van, hogy a közelítõ függvény valamely logaritmus függvény!

>

Ha meg kell állapítanunk az értelmezési tartományt, akkor a következõ dolgok némelyikét kell -sok esetben- sorra vennünk:

a) A tört nevezõje nem lehet 0;

b) Páros kitevõjû gyöke csak nem -negatív valós számnak van;

c) logaritmusa csak pozitív valós számnak van;

d) A $\frac{\pi}{k} + k\,\pi$ ($k \in Z$) alakú számoknak (szögeknek) nem értelmezhetõ a tangense;

e) A $k\,\pi$ ($k \in Z$) alakú számoknak (szögeknek) nem értelmezhetõ a kotangense.

Példa.

Határozzuk meg a valós számoknak azt a legbõvebb részhalmazát, amelyen az

$$f(x) = \sqrt{\log_{\frac{1}{2}}(3\,x - 4)}$$

utasítással függvény értelmezhetõ!

Megoldás.

A négyzetgyök függvény nem-negatív számokon értelmezhetõ, ezért

$$0 \le \log_{\frac{1}{2}}(3\,x - 4)$$

Az $\log_{\frac{1}{2}}(x)$ függvény a pozitív valós számok halmazám értelmezhetõ és szigorúan monoton csökkenõ, zérushelye az $x = 1$. Ezért

$$0 < 3\,x - 4 \text{ és } 3\,x - 4 \le 1$$

Oldjuk meg az egyenlõtlenségrendszert:

```
> solve({0 < 3*x-4,3*x-4 <= 1});
```

$$\left\{\frac{4}{3} < x, x \le \frac{5}{3}\right\} \qquad (1.1.4)$$

Tehát a $\frac{4}{3}$-nál nagyobb és $\frac{5}{3}$-nál nem nagyobb számok alkotják

Példa

Határozzuk meg a valós számoknak azt a legbõvebb részhalmazát, amelyen az

$$f(x) = \log_{x+2}\left(x^2 - 8\,x + 15\right)$$

utasítással függvény értelmezhetõ!

Megoldás

Két megszorítást kell tennünk:
1. A logaritmus alapja 1-tõl különbözõ poitív szám;
2. A logaritmus függvény értelmezési tartománya a pozitív számok halmaza.

Az 1. alapján

$$0 < x + 2,\ \ x + 2 \ne 1$$

A 2. alapján

$$0 < x^2 - 8\,x + 15$$

Oldjuk meg mindenekelõtt a két egyenlõtlenségbõl álló rendszert. Ezt a solve utasítással tehetjük meg.

```
> solve({0 < x+2,0 < x^2-8*x+15},x);
```

$$\{-2 < x, x < 3\},\ \{5 < x\} \qquad (1.1.5)$$

Az $x + 2 \ne 1$ egyenértékû az $x \ne -1$ feltétellel. Az értelmezési tartomány tehát:

$$?, x \ne -1$$

A szóbanforgó kifejezéseket és $y = 1$ egyenest jellemzõ intervallumon ábrázolva grafikus megerõsítést nyerhetünk.

```
> plot([x+2,x^2-8*x+15,1],x=-4..6,-4..10,color=[blue,red,green])
```

;



Zérushely

A függvény viselkedésének leírásában a jellegzetes pontok megtalálása sokat segíthet.A zérushelyek ismerete más tulajdonsággal, például a folytonossággal együtt különösen értékes.

Példa.

Határozzuk meg az $f(x) = x^6 - 6x^5 + 50x^3 - 45x^2 - 108x + 108$ *zérushelyeit. Jellemezzük a függvény viselkedését a zérushelyek környezetében!*

Megoldás.

A függvény szorzattá alakítását a factor utasítással kísérelhetjük meg.

```
> f:=x->x^6-6*x^5+50*x^3-45*x^2-108*x+108;
```

$$f := x \to x^6 - 6x^5 + 50x^3 - 45x^2 - 108x + 108 \qquad (1.2.1)$$

```
> factor(f(x));
```

$$(x-1)(x+2)^2(x-3)^3 \qquad (1.2.2)$$

A factor alkalmazása teljes síkerrek járt, ami nem meglepõ, hisz az f polinomfüggvény minden zérushelye racionális egész szám. a zérushelyek közvetlenül leolvashatók, a solve alkalmazása most inkább a megjelenítést szolgálja:

```
> zh:=[solve(f(x)=0)];
```

$$zh := [1, -2, -2, 3, 3, 3] \tag{1.2.3}$$

A solve a zérushelyeket multiplicitásuk ("többszörösség) szerint adja meg. Esetünkben az 1 egyszeres, a -2 kétszeres, a 3 pedig háromszoros zérushely. Vizsgáljuk meg, mit jelent ez a függvény viselkedése szempontjából! Mindenekelõtt rajzoljuk föl a függvény grafikonját a zérushelyeket tartalmazó intervallumban!

```
> plot(f(x),x=-2.5..4,color=blue,thickness=2);
```




```
>
```

A grafikon alapján úgy tûnik, hogy a függvény görbéjének a többszörös zérushelyeken érintõje az x tengely, azaz ezeken a helyeken a polinomfüggvény deriváltja is 0 értékû. Emellett azt is megfigyelhetjük, hogy a páratlan multiplicitású zérushelyen a függvény elõjelet vált, a páros multiplicitású zérushelyen ez nem következik be. Elsõ észrevételünket tétel formájában is megfogalmazzuk. (*Ez a tétel kiegészítõ ismeretet tartalmaz!!!*)

Tétel.

Ha egy α szám az f polinomnak pontosan k-szoros gyöke ($1 \le k$), akkor a polinom deriváltjának, a $D(f)$ polinomnak pontosan k-1-szeres gyöke. Speciálisan, ha α az f-nek egyszeres gyöke, akkor a polinom deriváltjának, $D(f)$-nek nem lehet gyöke.

Bizonyítás.

Legyen α pontosan k-szoros gyöke f-nek. Ekkor f az $x-\alpha$ gyöktényezõt pontosan a k-adik hatványon tartalmazza, azaz

$$f=(x-\alpha)^k g,$$

ahol már

$$g(\alpha)\neq 0,$$

ellenkezõ esetben ugyanis g osztható lenne $x-\alpha$-val, tehát f-nek α legalább $k+1$-szeres gyöke lenne, a feltevéssel ellentétben.Deriváljuk f-et

$$(\mathrm{D}(f))(x)=k\,(x-\alpha)^{(k-1)}g+(x-\alpha)^k Dd(g)=$$
$$(x-\alpha)^{(k-1)}\,(k\,g+(x-\alpha)\,\mathrm{D}(g)).$$

Az f deriváltjának ebbõl az alakjából nyilvánvaló, hogy α neki legalább $k-1$-szeres gyöke. Az is látszik, hogy α nem lehet $k-1$-nél nagyobb multiplicitású gyök, mert a második zárójelben $k\,g$ nem 0 az α helyen.

Paritás

A párosság-páratlanság vizsgálatakor sokszor célszerû elõször ellenpéldát keresni, azaz olyan x változóértéket, amelyre sem az

$$f(-x)=f(x)$$

sem az

$$f(-x)=-f(x)$$

egyenlõség nem teljesül.

Példa.

Mutassuk meg, hogy az

$$f(x)=\frac{x^5}{x-1}$$

függvény nem is páros és nem is páratlan. A függvény értelmezési tartománya legyen a valós számok megengedhetõ legbõvebb részhalmaza.

Megoldás.

A függvény értelmezési tartományához az $x=1$ hely nem tartozik hozzá, ugyanakkor az $x=-1$ helyen értelmezve van. Ezért nem lehet sem páros, sem páratlan.
Más esetben nem tekinthetünk el a definíciót követõ eljárás alkalmazásától.

Példa.

vizsgáljuk meg paritás szempontjából az

$$f(x)=\ln\left(x+\sqrt{x^2+1}\right)$$

függvényt!

Megoldás.

Láthatóan a függvény minden valós x-re értelmezhetõ. Számítsuk ki elõször a függvényt néhány helyen és az adott hely ellentett helyén:

```
> f:=x->ln(x+sqrt(x^2+1));
```

$$f:=x\rightarrow\ln\left(x+\sqrt{x^2+1}\right) \qquad (1.3.1)$$

```
> array(1..8,1..2,[['f(x)','f(-x)'],seq([evalf(f(-i)),evalf(f(i)
)],i=1..7)]);
```

$$\begin{bmatrix} f(x) & f(-x) \\ -.8813735879 & 0.8813735869 \\ -1.443635477 & 1.443635475 \\ -1.818446460 & 1.818446459 \\ -2.094712544 & 2.094712547 \\ -2.312438337 & 2.312438341 \\ -2.491779856 & 2.491779853 \\ -2.644120787 & 2.644120761 \end{bmatrix} \qquad (1.3.2)$$

Úgy látszik , hogy a függvény páratlan. Próbáljuk meg ezt igazolni!
Mutassuk meg, hogy

$$f(-x) = -f(x)$$

azaz

$$f(-x) + f(x) = 0$$

teljesül, az x választásától függetlenül.

```
> kif:=f(-x)+f(x);
```

$$kif := \ln\left(-x+\sqrt{x^2+1}\right) + \ln\left(x+\sqrt{x^2+1}\right) \qquad (1.3.3)$$

Alkalmazzuk a logaritmus mûveleti azonosságát, amely szerint:

$$\log(u) + \log(v) = \log(u\, v),$$

hacsak $0 < u$ és $0 < v$.

```
> combine(kif);
```

$$\ln\left(-x+\sqrt{x^2+1}\right) + \ln\left(x+\sqrt{x^2+1}\right) \qquad (1.3.4)$$

```
> assume(x,real);about(x);
Originally x, renamed x~:
  is assumed to be: real
```

```
> combine(kif);
```

$$\ln\left(\left(-x\sim+\sqrt{x\sim^2+1}\right)\left(x\sim+\sqrt{x\sim^2+1}\right)\right) \qquad (1.3.5)$$

Egyszerûsítsük a kifejezést:

```
> simplify(%);
```

$$0 \qquad (1.3.6)$$

```
>
```

A függvény valóban páratlan. Ezt mutatja a függvény képe is:

```
> plot(f(x),x=-10..10,color=blue,thickness=2);
```




```
>
```

Periódus

A periódus vizsgálatakor sokszor az ismert alapfügvény periódusból indulunk ki, és figyelembe vesszük, hogy az milyen átalakításokon esett át.

Példa.

Határozzuk meg az

$$f(x) = \cos(x)^2$$

függvény periódusát!

Megoldás.

A függvény cos függvény lineáris transzformáltjaként adható meg:

```
> x:='x':
> cos(x)^2=combine(cos(x)^2);
```

$$\cos(x)^2 = \frac{1}{2}\cos(2x) + \frac{1}{2} \tag{1.4.1}$$

Mivel a $\cos(x)$ periódusa 2π, a $\cos(x)^2$ periódusa $\frac{2\pi}{2} = \pi$.Ábrázoljuk a függvényt a $[0, 2\pi]$ intervallumon:

```
> plot(cos(x)^2,x=0..2*Pi,scaling=constrained,title=`A cos(x)^2
  periódusa Pi`);
```



Példa.

Legyen az f függvény értelmezve az $[0, p]$ intervallumon. Adjuk meg az f periodikus kiterjesztését, vagyis azt a g függvényt, amely az $[0, p]$ intervalumon azonos az f függvénnyel és amelyre:

$$g(x) = g(x + p)$$

teljesül minden x-re.

.

Megoldás.

Legyen a $p = 5$ és az f függvény:

$$f(x) = \begin{cases} x & -x \le 0 \textbf{ and } x < \frac{5}{2} \\ 5 - x & \frac{5}{2} - x \le 0 \textbf{ and } x - 5 \le 0 \end{cases}$$

Az f függvényt megadhatjuk a piecewise utasítással...

```
> p:=5:
> f:=x->piecewise(x>=0 and x<p/2,x,x>=p/2 and x<=p,5-x);
```

$$f := x \rightarrow piecewise\left(0 \le x \textbf{ and } x < \frac{1}{2}\,p, x, \frac{1}{2}\,p \le x \textbf{ and } x \le p, 5 - x\right) \quad (1.4.2)$$

```
> 'f(x)'=f(x);
```

$$f(x) = \begin{cases} x & 0 \le x \textbf{ and } x < \frac{5}{2} \\ 5 - x & \frac{5}{2} \le x \textbf{ and } x \le 5 \end{cases} \quad (1.4.3)$$

```
> 'f(4)'=f(4);
```

$$f(4) = 1 \quad (1.4.4)$$

...vagy eljárással is.

```
> f:=proc(x)
>  if x>=0 and x<p/2 then x elif x>=p/2 and x<=p then 5-x fi;
> end;
```

$f :=$ **proc**(x) (1.4.5)

if $0 <= x$ **and** $x < 1/2 * p$ **then**

x

elif $1/2 * p <= x$ **and** $x <= p$ **then**

$5 - x$

end if

end proc

```
> plot('f(x)',x=-1..6,color=blue);
```



A periódikus függvényt létrehozó eljárás:

```
> g:=proc(x)
>  local c:
>   if x<0 then c:=p*(frac(x/p)+1) else c:=p*(frac(x/p)) fi;
>   f(evalf(c));
> end;
```

$g :=$ **proc**(x) (1.4.6)
  local c;
  if $x < 0$ **then**
    $c := p * (\mathrm{frac}(x/p) + 1)$
  else
    $c := p * \mathrm{frac}(x/p)$
  end if;
  $f(evalf(c))$
end proc

Az imént kiszámítottuk az $f(4)$ értékét. Számítsuk ki a g értékét néhány

$$4 + 5\,k$$

alakú helyen, ahol k egész:

```
> seq('g'(4+5*k)=g(4+5*k),k=-5..5);
```

$g(-21)=1., g(-16)=1., g(-11)=1., g(-6)=1., g(-1)=1., g(4)=1., g(9)=1., g(14)=1., g(19)=1., g(24)=1., g(29)=1.$ (1.4.7)

Az értékek megfelelnek annak, hogy a g periódusa 5. Ábrázoljuk is a g függvényt :

```
> plot(g,-10..10,scaling=constrained,color=blue,thickness=2,
  discont=true,title=`Az f periodikus kiterjesztettje`);
```




```
>
```

Határérték, folytonosság

A függvény határértékét az x_0 helyen meg kell vizsgálnunk, ha a függvény az x_0 helyen nem folytonos, de az x_0-nak legalább egyik oldali környezetében értelmezve van. Ezen kívül megvizsgáljuk a függvény ∞-ben illetve $-\infty$-ben vett határértékét, ha a függvény elég nagy illetve elég kis számokra értelmezve van. A szakadási helyek felkutatására és a folytonosság vizsgálatára használhatjuk a Maple eljárásait is.

Példa.

Mutassuk meg, hogy az

$$f(x) = \frac{x}{x^2+1}$$

függvény folytonos.

Megoldás.

Mivel a függvény minden valós x értékre értelmezhetõ, azt kell megmutatnunk, hogy minden valós x-re folytonos is. Ez teljesül, mert folytonos függvények hányadosa és a nevezõ semmilyen valós x értékre sem 0. Megállapításunkat a Maple iscont ("folytonos-e") eljárása is megerõsíti:

```
> f:=x->1/(x^2+2);
```

$$f := x \to \frac{1}{x^2+2} \tag{1.5.1}$$

```
> iscont(f(x),x=-infinity..infinity);
```

$$true \tag{1.5.2}$$

Példa.

Vizsgáljuk meg folytonosság szempontjából az

$$f(x) = \frac{\cos(x)^2}{\sin(x) - \cos(2x)}$$

függvényt!

Megoldás.

Az f függvény folytonos függvények hányadosa. Tehát mindenütt folytonos, kivéve a nevezõ zérushelyeit.

```
> f:=x->cos(x)^2/(sin(x)-cos(2*x));
```

$$f := x \to \frac{\cos(x)^2}{\sin(x) - \cos(2x)} \tag{1.5.3}$$

Keressük meg a nevezõ zérushelyeit.

```
> zh:=[solve(denom(f(x)))];
```

$$zh := \left[-\frac{1}{2}\pi, \frac{1}{6}\pi, \frac{5}{6}\pi \right] \tag{1.5.4}$$

A függvény 2π szerint periodikus, ezek a nevezõ egy periódushoz tartozó zérushelyei, s egyszeresmind a függvény szakadási helyei. Vizsgáljuk meg a függvény határértékét a szakadási helyeken!

```
> for i to nops(zh) do
>  Limit(f(x),x=zh[i])=limit(f(x),x=zh[i]);
> od;
```

$$\lim_{x \to -\frac{1}{2}\pi} \left(\frac{\cos(x)^2}{\sin(x) - \cos(2x)} \right) = \frac{-2}{3}$$

$$\lim_{x \to \frac{1}{6}\pi} \left(\frac{\cos(x)^2}{\sin(x) - \cos(2x)} \right) = undefined$$

$$\lim_{x \to \frac{5}{6}\pi} \left(\frac{\cos(x)^2}{\sin(x) - \cos(2x)} \right) = undefined \tag{1.5.5}$$

```
>
```

Az $x = -\frac{\pi}{2}$ helyen van véges határérték, itt megszüntethetõ a szakadás a másik két helyen nincs határérték, ezeken a helyeken nem megszüntethetõ szakadás van. A kapott eredményeket segíthet értelmeznünk a számláló és a nevezõ együttes ábrája.

```
> plot([cos(x)^2,sin(x)-cos(2*x)],x=-Pi..2*Pi,thickness=2);
```



Láthatóan az $x = -\frac{\pi}{2}$ helyen (és a periodicitásnak megfelelõ helyeken) mind a számláló, mind a nevezõ nullává válik, mindkét függvény érinti az x tengelyt.
a többi helyen a számláló nem 0 értékû.A függvény képét is felrajzoljuk:

```
> plot(f(x),x=-2*Pi..2*Pi+1,-5..5,color=blue);
```



>

Ha a számlálóban a $\cos(x)^2$ helyett $\cos(x)$ szerepel, amely lassabban tart 0-hoz, midõn , akkor az $x = -\frac{\pi}{2}$ helyen is nem megszüntethetõ a szakadás!

```
> f1:=x->cos(x)/(sin(x)-cos(2*x));
```

$$f1 := x \rightarrow \frac{\cos(x)}{\sin(x) - \cos(2\,x)} \tag{1.5.6}$$

```
> limit(f1(x),x=-Pi/2);
```

$$\mathit{undefined} \tag{1.5.7}$$

```
> plot([numer(f1(x)),denom(f1(x)),f1(x)],x=-Pi/1-1..2*Pi,-5..5,
  color=[green,red,blue],thickness=[1,1,2],discont=true);
```



>

Deriválhatóság

Az f függvény az x_0 pontban akkor deriválható, ha az x_0 ponthoz tartozó különbségi hányadosának az x_0 pontban van (véges) határértéke, azaz
létezik a

$$\lim_{x \to x_0} \left(\frac{f(x) - f(x_0)}{x - x_0} \right)$$

véges határérték.
Azt is tudjuk, hogy ha az f függvény az x_0 pontban deriválható, akkor ott folytonos is.

Példa

Vizsgáljuk meg deriválhatóság szempontjából az

$$f(x) = |x^2 + x - 2|$$

függvényt!

Megoldás

```
> f:=x->abs(x^2+x-2);
```

$$f := x \to |x^2 + x - 2| \tag{1.6.1}$$

Tudjuk, hisz mintapéldaként szerepelt, hogy az abszolútérték függvény nem differenciálható az $x = 0$ helyen. Ezért valószínûsíthetjük, hogy az f függvény a zérushelyein nem deriválható! Keressük meg a zérushelyeket!

```
> zh:=[solve(f(x),x)];
```

$$zh := [-2, 1] \tag{1.6.2}$$

Ábrázoljuk az f függvényt a zérushelyeit tartalmazó intervallumban!

```
> plot(f(x),x=-3..2);
```



Az ábra is alátámasztja az iménti megjegyzésünket, vagyis azt , hogy a függvény az $x = -2$ és az $x = 1$ helyen nem differenciálható.. Adjuk meg szakaszonként értelmezett függvényként (piecewise) az f-et, Ezt a convert utasítással tehetjük meg. Ennek elsõ paramétere a konvertálandó függvényt definiáló kifejezés, a második pedig az elõállítandó típus, esetünkben a piecewise.

```
> convert(f(x),piecewise);
```

(1.6.3)

$$\begin{cases} x^2+x-2 & x \le -2 \\ -x^2-x+2 & x < 1 \\ x^2+x-2 & 1 \le x \end{cases} \tag{1.6.3}$$

Készítsük el a kapott, szakaszonként értelmezett függvény deriváltját!

```
> diff(%,x);
```

$$\begin{cases} 2x+1 & x < -2 \\ \mathit{undefined} & x = -2 \\ -2x-1 & x < 1 \\ \mathit{undefined} & x = 1 \\ 2x+1 & 1 < x \end{cases} \tag{1.6.4}$$

Látható, hogy a derivált nem definiált, tehát nem létezik az $x = -2$ és az $x = 1$ helyen.

```
> plot([f(x),D(f)(x)],x=-3..3,discont=true,color=[blue,red],
  thickness=2);
```



A discont eljárással is megállapíthatjuk a derivált szakadási helyeit.

```
> discont(D(f)(x),x);
```

$$\{-2, 1\} \tag{1.6.5}$$

Összefoglalva megállapíthatjuk, hogy az f függgvény az $x=-2$ és az $x=1$ helyeken nem differnciálható, másutt viszont igen. Mindez megfelel az elõzetes várakozásnak és a szemléletnek is. Az utóbbi azt jelenti, hogy amely helyeken "töréspontja" van a függvénynek, ott nem differenciálható.

Monotonitás, szélsõérték

A monotonítási intervallumok elkülönítését legtöbbször a derivált zérushelyeinek meghatározásával kezdjük Ezután a derivált zérushelyei által elkülönített intervallumokon megvizsgáljuk a derivált elõjelét. Azokon az intervallumokon, ahol a derivált pozitív, a függvény szigorúan monoton nõ, ahol a derivált negatív, ott a függvény szigorúan monoton csökken.
A helyi szélsõértékre vonatkozólag két elégséges feltételt ismertettünk.
Az elsõ szerint, ha $(\mathrm{D}(f))(x_0)=0$ és $\mathrm{D}(f)$ az x_0 pontban elõjelet vált, akkor az f függvénynek az x_0 helyen helyi szélsõértéke van.

A második elégséges feltétel szerint a $(\mathrm{D}(f))(x_0)=0$ és $\left(\left(\mathrm{D}^{(2)}\right)(f)\right)(x_0)\neq 0$ feltételek együttes teljesülése esetén az f függvénynek az x_0 pontban helyi szélsõértéke van.

Példa

Vizsgáljuk meg monotonitás és szélsõérték szempontjából az

$$f(x)=x^3-6x^2+9x-4$$

függvényt!

Megoldás

A függvény minden valós x értékre differenciálható. Ezért csak olyan pontokban lehet heyi szélsõértéke, ahol a derivált 0.
Vegyük föl mindenekelõtt a függvényt:

```
> f:=x->x^3-6*x^2+9*x-4;
```

$$f := x \to x^3-6x^2+9x-4 \tag{1.7.1}$$

Készítsük el a deriváltat!

```
> D(f)(x);
```

$$3x^2-12x+9 \tag{1.7.2}$$

Keressük meg a derivált zérushelyeit!

```
> zh:=[solve(D(f)(x))];
```

$$zh := [3, 1] \tag{1.7.3}$$

Most könnyû helyzeben vagyunk, mert a derivált másodfokú függvény, amelyet zérushelyeinek birtokában már egészen jól el tudunk képzelni. Ábrázoljuk a függvényt és deriváltját közös koordinátarendszerben!

```
> plot([f(x),D(f)(x),signum(D(f)(x))],x=-1..5,color=[blue,red,
  green],thickness=[2,2,1],discont=true,title=`f=kék,f'=piros,
  f''=zöld`);
```



>

Feltüntettük a derivált elõjelét is. Az ábra alapján, figyelembe véve a derivált zérushelyeire kapott értékeket, már mindent tudunk az f függvény monotonitásáról:
A függvény szigorúan monoton növekvõ a $(-\infty, 1)$ és a $(3, \infty)$ intervallumokon, és szigorúan monoton csökkenõ az (1,3) számközön. Ennek megfelelõen az 1 helyen helyi maximuma, a 3 helyen helyi minimuma van a függvénynek. Ezek értéke:

```
> f(1);
```

$$0 \tag{1.7.4}$$

```
> f(3);
```

$$-4 \tag{1.7.5}$$

Mi a helyzet azonban akkor, ha az f függvény az x_0 helyen nem differenciálható? Tekintsük a következõ függvényt!

Példa.

Vizsgáljuk meg monotonitás és szélsõérték szempontjából az

$$f(x) = \begin{cases} (2-x)^{\left(\frac{2}{3}\right)} & x \le 2 \\ (-2+x)^{\left(\frac{2}{3}\right)} & 2 < x \end{cases}$$

függvényt!

Megjegyzés: A függvény $f(x) = \left((x-2)^2\right)^{\left(\frac{1}{3}\right)}$ alakban is felírható.

Megoldás.

```
> f:=x->piecewise(x<=2,(2-x)^(2/3),x>2,(x-2)^(2/3));
```

$$f := x \rightarrow piecewise\left(x \le 2,\ (2-x)^{(2/3)},\ 2 < x,\ (x-2)^{(2/3)}\right) \qquad (1.7.6)$$

```
> f(2);
```

$$0 \qquad (1.7.7)$$

```
> 'f(x)'=f(x);
```

$$f(x) = \begin{cases} (2-x)^{(2/3)} & x \le 2 \\ (-2+x)^{(2/3)} & 2 < x \end{cases} \qquad (1.7.8)$$

A függvény zérushelye és "kritikus helye" egyuttal az $x = 2$. Készítsünk mindenekelõtt ábrát a függvényrõl az $x = 2$ helyrt tartalmazó intervallumon!

```
> plot(f(x),x=-2..6,color=blue,thickness=2,numpoints=1000,
  scaling=constrained);
```



Az $x = 2$ helyen a függvény folytonos, de az ábra szerint nem deriválható! Valóban a deriváltfüggvény

```
> 'D(f)(x)'=D(f)(x);
```

$$(\mathrm{D}(f))(x) = \begin{cases} -\frac{2}{3}\,\frac{1}{(2-x)^{(1/3)}} & x < 2 \\ \mathit{undefined} & x = 2 \\ \frac{2}{3}\,\frac{1}{(-2+x)^{(1/3)}} & 2 < x \end{cases} \tag{1.7.9}$$

Az $x = 2$ helyen a függvény nem differenciálható! A derivált határértékére:

```
> limit('D(f)(x)',x=2,left):=Limit(D(f)(x),x=2,left)=limit(D(f)
  (x),x=2,left);
```

(1.7.10)

$$\lim_{x \to 2-} (D(f))(x) := \lim_{x \to 2-} \left(\begin{cases} -\frac{2}{3} \frac{1}{(2-x)^{(1/3)}} & x < 2 \\ \mathit{undefined} & x = 2 \\ \frac{2}{3} \frac{1}{(-2+x)^{(1/3)}} & 2 < x \end{cases} \right) = -\infty \qquad (1.7.10)$$

```
> limit('D(f)(x)',x=2,right):=Limit(D(f)(x),x=2,right)=limit(D
  (f)(x),x=2,right);
```

$$\lim_{x \to 2+} (D(f))(x) := \lim_{x \to 2+} \left(\begin{cases} -\frac{2}{3} \frac{1}{(2-x)^{(1/3)}} & x < 2 \\ \mathit{undefined} & x = 2 \\ \frac{2}{3} \frac{1}{(-2+x)^{(1/3)}} & 2 < x \end{cases} \right) = \infty \qquad (1.7.11)$$

```
>
```

A deriváltnak az $x=2$ helyen határértéke sincs. Az $x=2$-nél kisebb x értékekre a derivált negatív, a 2-nél nagyobb változóértékekre pozitív. Tehát a függvény a $(-\infty, 2)$ intervallumon szigorúan monoton csökkenõ, a $(2, \infty)$ intervallumon pedig szigorúan monoton növekvõ. Emellett az $x=2$ helyen folytonos, hisz határértéke és helyettesítési értéke megegyezik:

```
> Limit(f(x),x=2)=limit(f(x),x=2);
```

$$\lim_{x \to 2} \left(\begin{cases} (2-x)^{(2/3)} & x \le 2 \\ (-2+x)^{(2/3)} & 2 < x \end{cases} \right) = 0 \qquad (1.7.12)$$

```
> 'f(2)'=f(2);
```

$$f(2) = 0 \qquad (1.7.13)$$

```
> is(limit(f(x),x=2)=f(2));
```

$$\mathit{true} \qquad (1.7.14)$$

Ezért elmondhatjuk, hogy a függvénynek az $x=2$ helyen helyi és abszolút minimuma van. Megálapításainkat összegzi a következõ ábra.

```
> kep:=plot([f(x),D(f)(x)],x=-2..6,view=[-2..6,-4..4],color=
  [blue,red],thickness=[2,1],discont=true):
> vonal:=plottools[line]([2,-5],[2,5],color=green,linestyle=3,
  legend=["x=2"]):
> plots[display]({kep,vonal},title=`A függvény és deriváltja az
  x=2 környezetében`);
Warning, the following options are not used by plottools: [legend = ["x=2"]]
```




```
> 
```

Elõfordul, hogy a függvény a szélsõértékét az értelmezési tartomány végpontjában, vagy az értelmezési tartományt alkotó intervallumok valamelyikének végpontjában veszi föl. Ilyenkor ez nem helyi szélsõértékhely, s a derivált zérushelyei között nem is szerepelhet, hisz ezeken a helyeken a függvény nem differenciálható. Ilyen módon veszi föl például az $f(x)=\sqrt{x}$ a minimumát az $x=0$ helyen.

```
> plot([sqrt(x),diff(sqrt(x),x)],x=0..4,0..5);
```



>

A **"szélsõérték-gyanús" helyek** tahát általában a következõk:
1) A derivált zérushelyei;
2) Azok az értelmezési tartományhoz tartozó belsõ pontok, ahol a függvény nem differenciálható;
3) Az értelmezési tartomány végpontjai (ha ilyenek vannak) illetve az értelmezési tartományt alkotó intervallumok végpontjai.

Konvexitás, inflexió

A függvény konvexitása illetve konkávitása lényeges információt jelent a függvény által leírt folyamatról. Ha a differenciálható függvény görbéje alulról konvex egy adott intervallumon, akkor ezen az intervallumon a deriváltja növekvõ. Ha a függvény növekvõ, akkor növekedésének üteme egyre nagyobb, ha csökkenõ, akkor egyre lassabban csökken. Hasonlóan, ha az alulról konkáv függvény növekvõ, akkor növekedése lassuló, ha csökkenõ, akkor egyre gyorsabban csökken. Az eltérõ konvexitású ívdarabok találkozási pontja az inflexiós pont.
Ha a függvény legalább kétszer deriválható, akkor konvexitási intervallumait a második derivált segítségével vizsgálhatjuk meg a legegyszerûbben. Ha az adott intervallumban a második derivált pozitív, s így az elsõ derivált növekvõ, akkok ebben az intervallumban a függvény görbéje alulról konvex, ha amásodik derivált negatív, akkor az elsõ derivált csökkenõ, s a függvénygörbe alulról

konkáv.

Példa

Vizsgáljuk meg konvexitás szempontjából az

$$f(x) = \ln\left(1 + x^2\right)$$

függvényt!

Megoldás

Vegyük föl mindenekelõtt a függvényt!

```
> f:=x->ln(1+x^2);
```

$$f := x \rightarrow \ln\left(x^2 + 1\right) \tag{1.8.1}$$

Képezzük a függvény második deriváltját!

```
> Diff(f(x),x$2)=diff(f(x),x$2);
```

$$\frac{d^2}{dx^2} \ln\left(x^2 + 1\right) = \frac{2}{x^2 + 1} - \frac{4 x^2}{\left(x^2 + 1\right)^2} \tag{1.8.2}$$

Hozzuk a deriváltat egyszerûbb alakra!

```
> Diff(f(x),x$2)=simplify(diff(f(x),x$2));
```

$$\frac{d^2}{dx^2} \ln\left(x^2 + 1\right) = -\frac{2\left(x^2 - 1\right)}{\left(x^2 + 1\right)^2} \tag{1.8.3}$$

Keressük meg a második derivált zérushelyeit!

```
> zh:=[solve(rhs(%),x)];
```

$$zh := [-1, 1] \tag{1.8.4}$$

Megjegyzés: A rhs(%) az elõzõ utasítás jobb oldalára való hivatkozást jelenti.
Rendezzük növekvõleg a zérushelyeket!

```
> zh:=sort(zh);
```

$$zh := [-1, 1] \tag{1.8.5}$$

Meg kell állapítanunk a második derivált elõjelét a zérushelyek által meghatározort intervallumokban! Ehhez tk. csak az elõjelre hatással lenni tudó $1 - x^2$ kifejezést kell figyelembe vennünk. De a függvény és a második derivált együttes ábrázolása mindent megmutat!

```
> plot([f(x),(D@@2)(f)(x)],x=-5..5,color=[blue,red],title=`f=
  kék,f''=piros`,thickness=2);
```



>

Innen már minden látszik. A $(-\infty, -1)$ és az $(1, \infty)$ intervallumban a második derivált negatív, ezért itt a függvény konkáv, a $(-1, 1)$ intervallumban a második deivált pozitív, ezért itt a függvény konvex. Az $x=-1$ és az $x=1$ inflexiós pontok.

Értékkészlet

A függvényvizsgálat utolsó elõtti állomása az értékkészlet megállapítása. A határértékvizsgálat, a szélsõértékek megállapítása után elegendõ információval rendelkezünk arra, hogy megadjuk a függvény által felvett értékek halmazát.

A függvény grafikonja

Utolsó lépésként megrajzoljuk a függvény grafikonját, vagy annak jellemzõ részletét. A Maple segítségével akkor juthatunk valóban jellemzõ grafikonhoz, ha az elõzõ vizsgálatok eredményeit figyelembe vesszük.Ez azt jelenti, hogy meg kell találnunk azt az alkalmas intervallumot, amelyben ábrázolva a függvényt, azt jól leíró grafikont kapunk.

Teljes függvényvizsgálat

Végezzük el - mintaképpen - két függvény teljes vizsgálatát!

Példa.

Vizsgáljuk meg az $f(x) = \frac{x^2}{(x-1)^2}$ *függvényt!*

Megoldás.

```
> f:=x->x^2/(x-1)^2;
```

$$f := x \to \frac{x^2}{(x-1)^2} \tag{2.1}$$

1. A függvény **értelmezési tartomány**a:
Az f függvény racionális törtfüggvény, tehát mindenütt értelmezett, kivéve a nevezõ zérushelyeit.
Az értelmezési tartomány tehát: minden valós szám, kivéve az $x = 1$ értéket.
2. Meghatározzuk a függvény **zérushelyeit :**

```
> `Zérushely`=solve(f(x)=0);
```

$$Zérushely = (0, 0) \tag{2.2}$$

Az f függvénynek egyetlen, kétszeres, zérushelye van az $x = 0$.
3. **Paritás vizsgálata:** Azértelmezési tartományból következik, hogy a függvény nem páros és nem páratlan.
4. A függvény láthatóan nem periodikus.
5 . **Határérték, folytonosság, deriválhatóság vizsgálata**:
A függvényt határérték szempontjából a szakadási helyein és (ha ennek van értelme) a ∞-ben és $-\infty$-ben kell megvizsgálni. A függvény az $x = 1$ hely kivételével mindenütt folytonos, sõt differenciálható, hiszen deriválható függvények hányadosa:

```
> Limit(f(x),x=1)=limit(f(x),x=1);
```

$$\lim_{x\to 1} \left(\frac{x^2}{(x-1)^2}\right) = \infty \tag{2.3}$$

Az $x = 1$ helyen a függvénynek nem megszüntethetõ szakadása van.

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x\to \infty} \left(\frac{x^2}{(x-1)^2}\right) = 1 \tag{2.4}$$

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x\to -\infty} \left(\frac{x^2}{(x-1)^2}\right) = 1 \tag{2.5}$$

Mindkét végtelenben 1 a határértéke a függvénynek. Ez eguttal azz is jelenti, hogy a függvény "vizszintes" aszimptotája, aszimptotafüggvénye az $y = 1$ egyenes.

6. **Monotonitás és szélsõérték vizsgálata**:
Az elsõ derivált zérushelyei a lehetséges szélsõértékhelyek:

```
> Diff(f(x),x)=diff(f(x),x);
```

$$\frac{d}{dx}\left(\frac{x^2}{(x-1)^2}\right) = \frac{2x}{(x-1)^2} - \frac{2x^2}{(x-1)^3} \tag{2.6}$$

Hozzuk egyszerûbb alakra a deriváltat:

```
> Diff(f(x),x)=normal(diff(f(x),x));
```

(2.7)

$$\frac{d}{dx}\left(\frac{x^2}{(x-1)^2}\right)=-\frac{2x}{(x-1)^3} \qquad (2.7)$$

Innen azonnal látszik, hogy a derivált az $x=0$ helyen 0 .A derivált elõjele is azonnal adódik, ha számlálójának és nevezõjének elõjelét figyelembe vesszük:
A derivált negatív x értékekre és 1-nél nagyobb x értékekre negatív, a (0,1) nyílt intervalumon pozitív. Tudjuk, hogy azokon az intervallumokon, ahol a derivált negatív a függvény szigorúan monoton csökkenõ, ahol a derivált pozitív, ott szigorúan monoton növekvõ. Tehát:

x	$(-\infty, 0)$	0	(0,1)	1	$(1, \infty)$
$\frac{\partial}{\partial x} f$	negatív	0	pozitív	-	negatív
f	csökkenõ	min	növekvõ	-	csökkenõ.

A helyi szélsõérték értéke $f(0)=0$

7. **Konvexitás, inflexiós pont.**
A második derivált, rögtön egyszerûbb alakban fölírva:

```
> Diff(f(x),x$2)=normal(diff(f(x),x$2));
```

$$\frac{d^2}{dx^2}\left(\frac{x^2}{(x-1)^2}\right)=\frac{2(2x+1)}{(x-1)^4} \qquad (2.8)$$

Ebbõl rögtön látszik, hogy az $x=-\frac{1}{2}$ helyen 0, ennél kisebb értékekre negatív, ennél nagyobb és az 1-tõl különbözõ értékekre pozitív. Tehát az $x=-\frac{1}{2}$ helyen a függvénynek inflexiós pontja van, ennél kisebb értékekre konkáv a függvény egyébként konvex. Részletezve:

x	$(-\infty, -\frac{1}{2})$	$-\frac{1}{2}$	$(-\frac{1}{2}, 1)$	$(1, \infty)$
$\frac{\partial^2}{\partial x^2} f$	negatív	0	pozitív	pozitív
f	konkáv	inflexió	konvex	konvex

8. A függvény értékkészlete: A monotonitás és a szélsõérték valamint a határértékrõl tett megállapításokat figyelembe véve adódik, hogy a függvény értékkészlete a nemnegatív számok halmaza.
9. Végül elkészítjük a **függvény grafikonját.** Ez, ha Maple nélkül csináljuk, komoly figyelmet követel. Figyelembe kell venni -egymással összevetve - valamennyi megállapításunkat és ki kell számítani néhány jellemzõ függvényértéket is.

```
> kep:=plot([f(x),1],x=-5..5,-0.5..10,color=[blue,cyan],thickness=
  [2,1],discont=true):
> vonal:=plottools[line]([1,-0.5],[1,10],linestyle=3,color=
  magenta):
> plots[display]({kep,vonal},title=`A függvény és aszimptotái`);
```



Példa.

Vizsgáljuk az $f(x) = (x+1)^2 \arctan\left(\frac{1}{x}\right)$ *függvényt!*

Megoldás.

Ezuttal a függvényvizsgálat szempontrendszerét kevésbé szorosan követve igyekszünk a függvényt megismerni. Bár nem minden ponton tudunk a függvényrõl teljesen bizonyosat mondani, az eredmények alapján mégis megbízhatóan tudjuk a függvényt jellemezni.

```
> restart:
> f:=x->(x+1)^2*arctan(1/x);
```

$$f := x \rightarrow (x+1)^2 \arctan\left(\frac{1}{x}\right) \quad (2.9)$$

Igencsak bonyolult a függvény! Azt mindenesetre látjuk, hogy a 0 kivételével minden x re értelmezhetõ. A zérushely láthatóan az $x = 1$.
A függvényt alkotó szorzat elsõ tényezõje ismert függvény, vizsgáljuk meg a második tényezõt!

```
> plot([arctan(1/x),arctan(x)],x=-10..10,discont=true,color=[red,
  blue],thickness=[2,1],legend=["arctan(1/x)","arctan(x)"]);
```



Meg kell vizsgálnunk a szakadási hely - tehát a 0 - környezetében és ∞-ben a függvény viselkedését!

```
> Limit(arctan(1/x),x=0,right)=limit(arctan(1/x),x=0,right);
```

$$\lim_{x \to 0+} \arctan\left(\frac{1}{x}\right) = \frac{1}{2}\pi \tag{2.10}$$

```
> Limit(arctan(1/x),x=infinity)=limit(arctan(1/x),x=infinity);
```

$$\lim_{x \to \infty} \arctan\left(\frac{1}{x}\right) = 0 \tag{2.11}$$

Érthetõ,hiszen $\lim_{x \to 0+} \left(\frac{1}{x}\right) = \infty$ és $\lim_{x \to \infty} \arctan(x) = \frac{\pi}{2}$, és $\lim_{x \to \infty} \left(\frac{1}{x}\right) = 0$, valamint $\lim_{x \to 0} \arctan(x) = 0$. A függvény páratlan, azaz $f(-x) = f(x)$ minden $x \neq 0$ esetén, ezért elég közvetlenül a fenti két határértéket vizsgálnunk.
Ábrázoljuk a függvény tényezõit közös koordinátarendszerben:

```
> plot({arctan(1/x),(x+1)^2},x=-3..3,discont=true);
```



A függvény két tényezõjét együtt szemlélve, s az elõbbi határérték-vizsgálat miatt látszik, hogy $\lim_{x\to\infty} f(x)$ és $\lim_{x\to-\infty} f(x)$ tisztázandó! A két tényezõ egyike ∞-be a másika 0-hoz tart, midõn .

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x\to\infty}\left((x+1)^2\arctan\left(\frac{1}{x}\right)\right)=\infty \tag{2.12}$$

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x\to-\infty}\left((x+1)^2\arctan\left(\frac{1}{x}\right)\right)=-\infty \tag{2.13}$$

A függvény deriválható függvényekbõl van összetéve, tehát minden 0-tól különbözõ x-re differenciálható:

```
> diff(f(x),x);
```

$$2\ (x+1)\arctan\left(\frac{1}{x}\right)-\frac{(x+1)^2}{x^2\left(1+\frac{1}{x^2}\right)} \tag{2.14}$$

Hozzuk a kifejezést egyszerûbb alakra! (A **simplify** hatása is ugyanez lenne. Próbáljuk ki!)

```
> normal(%);
```

$$\frac{(x+1)\left(2\arctan\left(\frac{1}{x}\right)x^2+2\arctan\left(\frac{1}{x}\right)-x-1\right)}{x^2+1} \tag{2.15}$$

Készítsünk a kifejezésből függvényt!

```
> fd1:=unapply(%,x);
```

$$fd1 := x \rightarrow \frac{(x+1)\left(2\arctan\left(\frac{1}{x}\right)x^2+2\arctan\left(\frac{1}{x}\right)-x-1\right)}{x^2+1} \tag{2.16}$$

Készítsük el a második deriváltat, mint kifejezést, majd hozzuk létre a derivált-függvényt!

```
> simplify(diff(fd1(x),x));
```

$$\frac{2\left(\arctan\left(\frac{1}{x}\right)x^4+2\arctan\left(\frac{1}{x}\right)x^2+\arctan\left(\frac{1}{x}\right)-x^3-x-2\right)}{\left(x^2+1\right)^2} \tag{2.17}$$

```
> fd2:=unapply(%,x);
>
```

$$fd2 := x \rightarrow \frac{2\left(\arctan\left(\frac{1}{x}\right)x^4+2\arctan\left(\frac{1}{x}\right)x^2+\arctan\left(\frac{1}{x}\right)-x^3-x-2\right)}{\left(x^2+1\right)^2} \tag{2.18}$$

A deriváltak elég bizarr kinézetûek, mielõtt az esetleges zérushelyeket megpróbálnánk megkeresnénk, ábrázoljuk a függvényt, és deriváltjait.

```
> plot([f(x),fd1(x),fd2(x)],x=-4..3,discont=true,color=[blue,red,
  green],legend=["f","D(f)","(D@@2)(f)"],thickness=[1,2,2]);
```



>

Az elsõ deriváltnak az x = -1 helyen levõ zérushelyét a képletbõl is láttuk, úgy tûnik több zérushelye nincs, de ha a ∞-ben és $-\infty$-ben vett határérték pozitív, ahogy az ábráról látszik, akkor ez már hihetõbb, s ha még azt is belátjuk, hogy a második derivált állandó elõjelû, akkor bizonyosak lehetünk.

```
> Limit(fd1(x),x=infinity)=limit(fd1(x),x=infinity);
```

$$\lim_{x \to \infty} \left(\frac{(x+1)\left(2\arctan\left(\frac{1}{x}\right)x^2 + 2\arctan\left(\frac{1}{x}\right) - x - 1\right)}{x^2+1} \right) = 1 \quad (2.19)$$

```
> Limit(fd1(x),x=-infinity)=limit(fd1(x),x=-infinity);
```

$$\lim_{x \to -\infty} \left(\frac{(x+1)\left(2\arctan\left(\frac{1}{x}\right)x^2 + 2\arctan\left(\frac{1}{x}\right) - x - 1\right)}{x^2+1} \right) = 1 \quad (2.20)$$

Ábrázoljuk a második deriváltat különbözõ intervallumokon! A függvények együttes ábrájából nem látszott világosan, az második derivált viselkedése a 2 kezdetû intervallumokon. Nézzük:

```
> plot(fd2(x),x=2..10,discont=true);
```



Negatív x értékekre a második derivált viselkedése nehezen tanulmányozható, mindenesetre a $-\infty$-ben vett határérték:

```
> Limit(fd2(x),x=-infinity)=limit(fd2(x),x=-infinity);
```

$$\lim_{x\to-\infty}\left(\frac{2\left(\arctan\left(\frac{1}{x}\right)x^4+2\arctan\left(\frac{1}{x}\right)x^2+\arctan\left(\frac{1}{x}\right)-x^3-x-2\right)}{\left(x^2+1\right)^2}\right)=0 \qquad (2.21)$$

A második derivált egy negatív értékekbõl álló intervallumon:

```
> plot(fd2(x),x=-10..-5,discont=true);
```



Mindezekbõl igen valószínû, hogy a második derivált minden negatív x értékre negatív.

Az elsõ derivált egyetlen zérushelyét a fsolve is megadja!

```
> fsolve(fd1(x),x=-1..4);
```

$$-1. \tag{2.22}$$

A második derivált ábrázolása során adódott, hogy a [2,4] számközben van zérushelye a $\left(D^{(2)}\right)(f)$-nek:

```
> fsolve(f2(x),x=2..4);
```

$$fsolve(f2(x), x, 2..4) \tag{2.23}$$

Itt a második derivált elõjelet is vált, tehát inflexiós pontot találtunk!
Rajzoljuk föl most még egyszer a függvényt!

```
> plot(f(x),x=-4..3,discont=true,color=blue,title=`Az f(x) képe`,
  thickness=2);
```



>

Úgy tûnik, hogy egyenes a függvény aszimptotája? Mikor is mondjuk,hogy az $y = m\,x + b$ egyenes aszimptotája az f függvénynek?

DEFINÍCIÓ: Az $y = m\,x + b$ alakú egyenes az f függvény aszimptota egyenese, ha:

$$\lim_{x\to\infty}\left(\frac{f(x)}{x}\right) = m \quad \text{és} \quad \lim_{x\to\infty}\,(f(x) - m\,x) = b$$

teljesül.

Határozzuk meg az m és a b értékét!

```
> Limit(f(x)/x,x=infinity)=limit(f(x)/x,x=infinity);
```

$$\lim_{x\to\infty}\left(\frac{(x+1)^2\arctan\left(\frac{1}{x}\right)}{x}\right) = 1 \qquad (2.24)$$

```
> m:=rhs(%);
```

$$m := 1 \qquad (2.25)$$

```
> Limit(f(x)-m*x,x=infinity)=limit(f(x)-m*x,x=infinity);
```

$$\lim_{x\to\infty}\left((x+1)^2\arctan\left(\frac{1}{x}\right) - x\right) = 2 \qquad (2.26)$$

```
> b:=rhs(%);
```

$$b := 2 \tag{2.27}$$

Az ***aszimptota:***

```
> y=m*x+b;
```

$$y = x + 2 \tag{2.28}$$

```
> aszimptota:=unapply(rhs(%),x);
```

$$aszimptota := x \rightarrow x + 2 \tag{2.29}$$

```
>  plot([f(x),aszimptota(x)],x=-4..4,title=`A függvény és
aszimptotája`,color=[blue,red],thickness=[2,1],discont=true);
```




```
>
```

Az aszimptota meghatározása és ábrázolása megerõsíti eddigi vizsgálataink eredményeit!

Feladat.

A példát követve vizsgálja meg a következõ függvényeket:

$$f(x) = (x+1)^2 \arctan\left(\frac{1}{x^2}\right); \qquad f(x) = (x+1)^2 \arctan\left(\frac{1}{x^3}\right);$$

$$f(x) = (x+1)^3 \arctan\left(\frac{1}{x^2}\right).$$

>

Kérdések, feladatok

Ellenõrzõ kérdések

1. Melyek a függvényvizsgálat szempontjai?
2. Mir értünk a függvény értelmezési tartománya alatt?
3. Mit nevezünk a függvény zérushelyének?
4. Mikor mondjuk a függvényt párosnak (páratlannak)?
5. Mit értünk az alatt, hogy a függvény periódusa a $0 < p$ szám?
6. Mely helyeken kell a függvényt határérték szempontjából megvizsgálni?
7. Melyek a szélsõértékgyanús helyek?
8. Hogyan határozhatjuk meg a szélsõértékeket?
9. Mirõl nyújt információt a függvény konvexitása?
10. Tegyük föl, hogy az f függvény legalább kétszer differenciálható! Mely pontokban lehet a függvénynek inflexiója?
11. Milyen elégséges feltételét ismeri az inflexiós pont létezésének?

Gyakorló feladatok

Végezzünk teljes függvényvizsgálatot az alábbi függvényeken:

1.$f(x) = 3x - x^3$; 2. $f(x) = 1 + x^2 - \frac{x^4}{2}$; 3. $f(x) = (x+1)(x-2)^2$; 4. $f(x) = \frac{4x}{x^2+1}$; 5. $f(x) = \frac{2-x^2}{1+x^4}$; 6. $f(x) = \frac{x-2}{\sqrt{x^2+1}}$;

7. $f(x) = \sin(x) + \cos(x)^2$; 8. $f(x) = e^{(2x-x^2)}$; .9. $f(x) = \frac{e^x}{1+x}$; 10. $f(x) = \frac{\ln(x)}{\sqrt{x}}$; 11. $f(x) = x \arctan(x)$.